Economic equilibrium and optimization problems using GAMS
Notes 8: open (trading) economy models

James R. Markusen
University of Colorado, Boulder

Small open economy

Small open economy 2x2 (two goods, two factors) in which the rest of the world is not explicitly modeled.

Know as the “small-country assumption”: the country faces fixed world prices

Trading opportunities are summarized by simple functions which allow the economy to transform one good (an export) into another (an import).

The "technology" of these functions represents world prices. We will assume that these technologies or price ratios are fixed in this model.

Country exports X_1 in exchange for X_2 (activities E1, M2)

We will allow for the fact that some policy or endowment change could actually reverse the direction of trade by specifying (initially inactive) functions that transform goods in the opposite direction.

Initial data, in which 50 units of good X_1 are exchanged for 50 units of good X_2 at an implicit price ratio of $p_1/p_2 = 1$.

	Production Sectors					Consumer
Markets	X1	X2	E1	M2	W	CONS
P1	150		-50		-100	
P2		50		50	-100	
PL	-135	-5				140
PK	-15	-45				60
PW					200	-200
PFX			50	-50		

Technology parameters are specified in these functions that allow the modeler to change the terms of trade.

```
PARAMETERS
   PE2     Export price of good 2,
   PM1     Import price of good 1,
   PE1     Export price of good 1,
   PM2     Import price of good 2,
   TM2     Import tariff for good 2;

PE1 = 1;
PM2 = 1;
PE2 = 0.999;
PM1 = 1.001;
```

E1 and M2 production activities that are the initially active trade links.

Small trade costs on inactive links prevent model degeneracy and "round tripping".

If all prices = 1, then if E1 = M2 = 1 is a solution, then so is E1 = 2, M1 = 1, M2 = 1.

While we could specify this activity as directly transforming X_1 into X_2, in more complicated models it proves useful to define another good which we will call “foreign exchange” and whose price is denoted PFX.

All trade is mediated through the “foreign exchange market”.

Thus activity E1 transforms X_1 into foreign exchange and M2, the import activity for good 2, transforms foreign exchange into imports of good 2.

There are four trade activities as just noted.

```
PRF_E1..  0*P1 =G= 50*PFX*PE1;

PRF_E2..  50*P2 =G= 50*PFX*PE2;

PRF_M1..  50*PFX*PM1 =G= 50*P1;

PRF_M2..  50*PFX*PM2 =G= 50*P2;
```

The equation for “foreign exchange is the trade balance condition:

```
MKT_PFX.. 50*E2*PE2 + 50*E1*PE1 =G= 50*PM2*M2 + 50*PM1*M1;
```

```
$TITLE  M8-1.GMS: Small open economy 2x2
* strips out trade costs and tariffs for simplicity of exposition

$ONTEXT
 CALIBRATION: country exports X1, imports X2
 in free-trade SOE

                 Production Sectors                      Consumer

Markets   |     X1      X2      E1      M2      W       CONS
----------------------------------------------------------------
P1        |     150             -50             -100
P2        |             50              50      -100
PL        |    -135     -5                              140
PK        |     -15    -45                               60
PW        |                                  200     -200
PFX       |                      50     -50
----------------------------------------------------------------
$OFFTEXT

* the first four parameters allow changes in (exogenous) world prices


PARAMETERS
        PE2     Export price of good 2          /0.999/
        PM1     Import price of good 1          /1.001/
```

```
        PE1     Export price of good 1          /1/
        PM2     Import price of good 2          /1/;

NONNEGATIVE VARIABLES
        X1      Activity level for sector X1
        X2      Activity level for sector X2
        E1      Activity level for sector E1
        E2      Activity level for sector E2
        M1      Activity level for sector M1
        M2      Activity level for sector M2
        W       Activity level for sector W

        P1      Price index for commodity X
        P2      Price index for commodity Y
        PL      Price index for primary factor L
        PK      Price index for primary factor K
        PW      Price index for welfare (consumer price index)
        PFX     Real exchange rate index

        CONS    Income definition for CONS;

EQUATIONS
        PRF_X1  Zero profit for sector X1
        PRF_X2  Zero profit for sector X2
        PRF_E1  Zero profit for sector E1
        PRF_E2  Zero profit for sector E2
```

```
         PRF_M1  Zero profit for sector M1
         PRF_M2  Zero profit for sector M2
         PRF_W   Zero profit for sector W

         MKT_X1  Supply-demand balance for commodity X1
         MKT_X2  Supply-demand balance for commodity X2
         MKT_PFX Supply-demand balance for commodity PFX
         MKT_L   Supply-demand balance for primary factor L
         MKT_K   Supply-demand balance for primary factor L
         MKT_W   Supply-demand balance for aggregate demand

         I_CONS  Income definition for CONS;

*        Zero profit conditions

PRF_X1..    150*PL**(0.9) * PK**(0.1) =G= 150*P1;

PRF_X2..    50*PL**(0.1) * PK**(0.9) =G=  50*P2;

PRF_E1..    50*P1 =G= 50*PFX*PE1;

PRF_E2..    50*P2 =G= 50*PFX*PE2;

PRF_M1..    50*PFX*PM1 =G= 50*P1;

PRF_M2..    50*PFX*PM2 =G= 50*P2;
```

```
PRF_W..      100*P1**0.5 * P2**0.5 =G= 100*PW;

*       Market clearance conditions

MKT_X1..     150*X1 + 50*M1 =G= 50*E1 + 100*W*PW/P1;

MKT_X2..     50*X2  + 50*M2 =G= 50*E2 + 100*W*PW/P2 ;

MKT_PFX..    50*E2*PE2 + 50*E1*PE1 =G= 50*PM2*M2 + 50*PM1*M1;

MKT_W..      200*W =G= CONS / PW;

MKT_L..      140 =G= 135*X1 * P1/PL +  5*X2*P2/PL;

MKT_K..      60 =G=   15*X1*P1/PK + 45*X2*P2/PK;

*       Income balance

I_CONS..     CONS =E= 140*PL + 60*PK;

MODEL SOE1 /PRF_X1.X1, PRF_X2.X2, PRF_E1.E1, PRF_E2.E2,
            PRF_M1.M1, PRF_M2.M2, PRF_W.W,
            MKT_X1.P1, MKT_X2.P2, MKT_PFX.PFX, MKT_L.PL,
            MKT_K.PK, MKT_W.PW, I_CONS.CONS /;
```

```
*       set SOE values:

X1.L    =1;
X2.L    =1;
E2.L    =0;
M1.L    =0;
E1.L    =1;
M2.L    =1;
W.L     =1;

P1.L    =1;
P2.L    =1;
PFX.L   =1;
PK.L    =1;
PL.L    =1;
CONS.L  =200;

* choose the real consumer price index as numeraire

PW.FX   =1;

* check for calibration and starting-value errors

SOE1.ITERLIM = 0;
SOLVE SOE1 USING MCP;
SOE1.ITERLIM = 2000;
```

```
SOLVE SOE1 USING MCP;

* counterfactual: a terms-of-trade improvement

PE1 = 1.2;
PM1 = 1.21;
SOE1.ITERLIM = 2000;
SOLVE SOE1 USING MCP;
```

8.2 Small open economy: tariffs versus trade costs

Zero-profit conditions for trade activities

PFX - price of a unit of “foreign exchange”
TC - gross real trade cost: 1 + the trade-cost rate
TR - gross tariff: 1 + the ad valorem tariff rate

TC = TR = 1: costless trade: tariffs, trade costs *appear* symmetric

```
PRF_E1..    50 * P1 =G= 50 * PFX * PE1;

PRF_E2..    50 * P2 =G= 50 * PFX * PE2;

PRF_M1..    50 * PFX * PM1 * TC * TR =G= 50 * P1;

PRF_M2..    50 * PFX * PM2 * TC * TR =G= 50 * P2;
```

TC and TR appear the same way in the pricing equations. However, they are quite different from one another.

Modeling a tariff that is costlessly collected and redistributed to consumers lump sum.

Recall that TR is defined as the gross tariff (one plus the ad valorem rate), so (TR - 1) is the tariff rate. The tariff revenue from equation PRF_M1 is:

```
[50 * PFX * PM2] * (TR-1) * M2
```

The term in brackets [ ] is the cost of 50 units (the benchmark value) of imports of good 2, (TR - 1) is the tariff rate, and M2 is the activity level, calibrated at one initially.

The assumption that the tariff can be costlessly redistributed means that this amount must be entered in the consumer's budget constraint: Income = factor income + tariff revenue.

```
CONS =E= 140*PL + 60*PK + 50*PFX*PM2*(TR-1)*M2;
```

Modeling an "iceberg" trade cost. Assume that the trade cost is paid in units of the good itself.

This is sometimes referred to as "iceberg trade costs". Some of the good "melts" in transit and less arrives than is shipped.

Let p* and M be the price received by the exporter and M be the number of units shipped.

Then the price paid by the importer is (p*TC). The quantity received by the importer is (M/TC): the quantity that arrives unmelted.

Note that revenue received by the exporter and paid by the importer are the same: the payments do and must balance.

p*M = (p*TC)(M/TC)

The trade cost obviously does not generate any revenue and does not become a term in the consumer's budget constraint.

However, it reappears in the model in the supply = demand equations for the two goods: the amount entered is the amount received, not the amount shipped.

The market-clearing conditions for the two goods are production plus imports (received) minus exports equal domestic demand.

```
150*X1 + 50*M1/TC  =G=  50*E1 + 100*W*PW/P1

 50*X2 + 50*M2/TC  =G=  50*E2 + 100*W*PW/P2
```

Finally, there is a market-clearing condition for foreign exchange that determines it price (PFX):

earnings from exports = import costs

```
50*E2*PE2 + 50*E1*PE1 =G= 50*PM2*M2 + 50*PM1*M1;
```

Counter-factual: show the welfare effects of a tariff versus a real trade costs.

Two sets are defined:

I will loop over values of trade costs or tariffs
J will denote the tariff and trade cost scenarios respectively.

Then some two-dimensional parameters are defined in order to extract some results after solving.

The * in RESULTS(I, *) is called a “wild card” and its use should be obvious below.

```
SETS I indexes 25 different gross cost levels      /I1*I25/
     J indexes 2 scenarios: 1 = tar 2 = trade cost /J1*J2/;
```

```
PARAMETERS
     RATE(I)
     WELFARE(I,J)
     IMP2(I,J)
     TRATE(I,J)
     TCOST(I,J)
     RESULTS(I, *);
```

Then there is the double loop, the inner loop over I sets the cost rate, starting at 1.

For J = 1, the rate is a tariff, for J = 2, the rate is a trade costs.

```
LOOP(J,
LOOP(I,

TC = 1; TR = 1;
RATE(I) = 1 + .05*ORD(I) - 0.05;
TR$(ORD(J) EQ 1) = RATE(I);
TC$(ORD(J) EQ 2) = RATE(I);
```

The *left-hand* exception operator $

```
TR$(ORD(J) EQ 1) = RATE(I);
```

is read “set the parameter TR equal to RATE if the ordinal value of J is equal to one, otherwise leave TR unchanged from it’s existing value”.

For future reference, the meaning is different if the exception operator is on *the right-hand* side.

```
TR= RATE(I)$(ORD(J) EQ 1;
```

is read “set the tariff TR equal to RATE if the ordinal value of J is equal to one, otherwise set TR equal to *zero*”. This is something quite different!

The parameter IMP(I, J) extracts the equilibrium value of exports in the two scenarios.

```
SOLVE ALGEBRAIC USING MCP;

WELFARE(I,J) = W.L;
IMP2(I,J) = M2.L/TC;

);
);

RESULTS(I, "RATE") = RATE(I);
RESULTS(I, "WELTR") = WELFARE(I, "J1");
RESULTS(I, "WELTC") = WELFARE(I, "J2");
RESULTS(I, "IMP2TR")= IMP2(I,"J1");
RESULTS(I, "IMP2TC")= IMP2(I,"J2");

DISPLAY RESULTS;
```

```
$TITLE  M8-2.GMS: Small open economy 2x2
*       contrasts tariffs versus (iceberg) trade costs

$ONTEXT
 CALIBRATION: country exports X1, imports X2
 in free-trade benchmark

                 Production Sectors                          Consumer

Markets   |     X1      X2      E1      M2      W       CONS
----------------------------------------------------------------------
P1        |     150             -50             -100
P2        |             50              50      -100
PL        |    -135     -5                              140
PK        |     -15    -45                               60
PW        |                                  200     -200
PFX       |                     50     -50
----------------------------------------------------------------------
$OFFTEXT

* the first four parameters allow changes in (exogenous) world


PARAMETERS
        PE2     Export price of good 2          /0.999/
        PM1     Import price of good 1          /1.001/
```

```
        PE1     Export price of good 1          /1/
        PM2     Import price of good 2          /1/
        TR      Import tariff (gross basis)     /1/
        TC      Trade costs (gross basis)       /1/
        CONSX1  Consumption of X1
        CONSX2  Consumption of X2;


NONNEGATIVE VARIABLES
        X1      Activity level for sector X1
        X2      Activity level for sector X2
        E1      Activity level for sector E1
        E2      Activity level for sector E2
        M1      Activity level for sector M1
        M2      Activity level for sector M2
        W       Activity level for sector W

        P1      Price index for commodity X
        P2      Price index for commodity Y
        PL      Price index for primary factor L
        PK      Price index for primary factor K
        PW      Price index for welfare (consumer price index)
        PFX     Real exchange rate index

        CONS    Income definition for CONS;
```

```
EQUATIONS
        PRF_X1  Zero profit for sector X1
        PRF_X2  Zero profit for sector X2
        PRF_E1  Zero profit for sector E1
        PRF_E2  Zero profit for sector E2
        PRF_M1  Zero profit for sector M1
        PRF_M2  Zero profit for sector M2
        PRF_W   Zero profit for sector W

        MKT_X1  Supply-demand balance for commodity X1
        MKT_X2  Supply-demand balance for commodity X2
        MKT_PFX Supply-demand balance for commodity PFX
        MKT_L   Supply-demand balance for primary factor L
        MKT_K   Supply-demand balance for primary factor L
        MKT_W   Supply-demand balance for aggregate demand

        I_CONS  Income definition for CONS;

*       Zero profit conditions

PRF_X1..    150*PL**(0.9) * PK**(0.1) =G= 150*P1;

PRF_X2..    50*PL**(0.1) * PK**(0.9) =G=  50*P2;

PRF_E1..    50*P1 =G= 50*PFX*PE1;
```

```
PRF_E2..    50*P2 =G= 50*PFX*PE2;

PRF_M1..    50*PFX*PM1*TC*TR =G= 50*P1;

PRF_M2..    50*PFX*PM2*TC*TR =G= 50*P2;

PRF_W..     100*P1**0.5 * P2**0.5 =G= 100*PW;

*       Market clearance conditions

MKT_X1..    150*X1 + 50*M1/TC =G=  50*E1 + 100*W*PW/P1;

MKT_X2..    50*X2 + 50*M2/TC =G= 50*E2 + 100*W*PW/P2 ;

MKT_PFX..   50*E2*PE2 + 50*E1*PE1 =G= 50*PM2*M2 + 50*PM1*M1;

MKT_W..     200*W =G= CONS / PW;

MKT_L..     140 =G= 135*X1 * P1/PL +  5*X2*P2/PL;

MKT_K..     60 =G=   15*X1*P1/PK + 45*X2*P2/PK;

*       Income balance

I_CONS..    CONS =E= 140*PL + 60*PK + 50*PFX*PM2*(TR-1)* M2;
```

```
MODEL SOE /PRF_X1.X1, PRF_X2.X2, PRF_E1.E1, PRF_E2.E2,
                PRF_M1.M1, PRF_M2.M2, PRF_W.W,
                MKT_X1.P1, MKT_X2.P2, MKT_PFX.PFX, MKT_L.PL,
                MKT_K.PK, MKT_W.PW, I_CONS.CONS /;

*       set benchmark values:

X1.L    =1;
X2.L    =1;
E2.L    =0;
M1.L    =0;
E1.L    =1;
M2.L    =1;
W.L     =1;

P1.L    =1;
P2.L    =1;
PFX.L   =1;
PK.L    =1;
PL.L    =1;
CONS.L  =200;

* choose the real consumer price index as numeraire

PW.FX   =1;
```

```
* check for calibration and starting-value errors

SOE.ITERLIM = 0;
SOLVE SOE USING MCP;

SOE.ITERLIM = 2000;
SOLVE SOE USING MCP;



* SHOW HOW TO DO MULTIPLE SCENARIOS
* SHOW DIFFERENCE BETWEEN TARIFF AND TRADE COST OF EQUAL RATES

SETS I indexes 25 different gross cost levels          /I1*I25/
     J indexes 2 scenarios: 1 = tariff 2 = trade cost /J1*J2/;

PARAMETERS
 RATE(I)
 WELFARE(I,J)
 IMP2(I,J)
 TRATE(I,J)
 TCOST(I,J)
 RESULTS(I, *);

LOOP(J,
LOOP(I,
```

```
TC = 1; TR = 1;
RATE(I) = 1 + .05*ORD(I) - 0.05;
TR$(ORD(J) EQ 1) = RATE(I);
TC$(ORD(J) EQ 2) = RATE(I);

SOLVE SOE USING MCP;

WELFARE(I,J) = W.L;
IMP2(I,J) = M2.L/TC;

);
);

RESULTS(I, "RATE") = RATE(I);
RESULTS(I, "WELTR") = WELFARE(I, "J1");
RESULTS(I, "WELTC") = WELFARE(I, "J2");
RESULTS(I, "IMP2TR")= IMP2(I,"J1");
RESULTS(I, "IMP2TC")= IMP2(I,"J2");

DISPLAY RESULTS;


* Write parameter RESULTS to an Excel file M8.XLS,
* starting in Sheet1
```

```
Execute_Unload 'M8.gdx' RESULTS
execute 'gdxxrw.exe M8.gdx par=RESULTS rng=SHEET1!'

* simpler but less sophisticated dump to excel

$LIBINCLUDE XLDUMP RESULTS M8.XLS SHEET2!
```

8.3 Small open economy: calibrating to tariffs in the benchmark

This model has a 20% tariff in the benchmark data. It is important to keep track of prices and trade balance in this situation.

	Production Sectors					Consumer
Markets	X1	X2	E1	M2	W	CONS
P1	150		-50		-100	
P2		40		60	-100	
PL	-100	-20				120
PK	-50	-20				70
PW					200	-200
PFX			50	-50		
Tariff				-10		10

Trade balance holds, since exports of good 1 generate 50 units of foreign exchange and those 50 units are spent on imports.

Consumers spend 60 on imports of which 10 is the tariff, or 20% of the value of imports (10/50).

The convention we adopt here is that all domestic prices are equal to 1 initially. Since the export good X_1 is freely traded, then the international price of X_1 is also equal to 1.

However, the import good has a domestic price equal to (1 + TM2) times the world price, so if the domestic price is equal to 1 then the world price must equal 1/ (1 + TM2) = 1/1.2.

Thus the -50 in the column M2 of the matrix is interpreted as 60 units at a price of 1/1.2.

The counterfactual experiment is free trade.

```
$TITLE: M8-3: Small open economy with a benchmark tariff

$ONTEXT
In this example, units are chosen such that all DOMESTIC prices
equal one initially.  Implied world price of import good
X2: P2 = 1/1.2

                 Production Sectors                        Consumer

Markets       |     X1       X2       E1       M2       W       CONS
--------------------------------------------------------------------
P1            |    150                -50              -100
P2            |              40                 60     -100
PL            |   -100      -20                                 120
PK            |    -50      -20                                  70
PW            |                                     200      -200
PFX           |                          50     -50
TARIFF        |                                 -10              10
--------------------------------------------------------------------
$OFFTEXT

PARAMETERS
        PE2     Export price of good 2
        PM1     Import price of good 1
        PE1     Export price of good 1
        PM2     Import price of good 2
```

```
        TM2     Import tariff for good 2;

PE1     = 1;
PM2     = 1 / (1.2);
PE2     = PM2 * 0.99;
PM1     = 1.01;
TM2     = 0.2;


POSITIVE VARIABLES
        X1      Activity level for sector X1
        X2      Activity level for sector X2
        E1      Activity level for sector E1
        E2      Activity level for sector E2
        M1      Activity level for sector M1
        M2      Activity level for sector M2
        W       Activity level for sector W (Hicksian welfare index)
        P1      Price index for commodity X
        P2      Price index for commodity Y
        PL      Price index for primary factor L
        PK      Price index for primary factor K
        PW      Price index for welfare (expenditure function)
        PFX     Read exchange rate index
        CONS    Income definition for CONS;
```

```
EQUATIONS
        PRF_X1  Zero profit for sector X1
        PRF_X2  Zero profit for sector X2
        PRF_E1  Zero profit for sector E1
        PRF_E2  Zero profit for sector E2
        PRF_M1  Zero profit for sector M1
        PRF_M2  Zero profit for sector M2
        PRF_W   Zero profit for sector W (Hicksian welfare index)

        MKT_X1  Supply-demand balance for commodity X1
        MKT_X2  Supply-demand balance for commodity X2
        MKT_PFX Supply-demand balance for commodity PFX
        MKT_L   Supply-demand balance for primary factor L
        MKT_K   Supply-demand balance for primary factor L
        MKT_W   Supply-demand balance for aggregate demand

        I_CONS  Income definition for CONS;

*       Zero profit conditions

PRF_X1..        150 * PL**(2/3) * PK**(1/3) =G= 150 * P1;

PRF_X2..        40  * PL**(0.5) * PK**(0.5) =G= 40  * P2;

PRF_E1..        50 * P1 =G= 50 * PFX * PE1;
```

```
PRF_E2..        60 * P2 =G= 60 * PFX * PE2;

PRF_M1..        50 * PFX * PM1 =G= 50 * P1;

PRF_M2..        60 * PFX * PM2 *(1+TM2) =G= 60 * P2;

PRF_W..         200 * P1**0.5 * P2**0.5 =G= 200 * PW;

*       Market clearance conditions

MKT_X1..        150 * X1 + 50 * M1 =G=  50 * E1 + 100 * W *PW/P1;

MKT_X2..        40 * X2 + 60 * M2 =G= 60 * E2 + 100 *W *PW/P2 ;

MKT_PFX..       60 * E2 * PE2 + 50 * E1 * PE1 =G=
                60 * M2 * PM2 + 50 * PM1 * M1;

MKT_W..         200 * W =G= CONS / PW;

MKT_L..         120  =G= 100 * X1 * P1/PL + 20 * X2 * P2/PL;

MKT_K..         70 =G= 50 * X1 * P1/PK + 20 * X2 * P2/PK;

*       Income balance

I_CONS..        CONS =E= 120*PL + 70*PK + 60 * PFX * PM2 * M2 * TM2;
```

```
MODEL SOETARIFF /PRF_X1.X1, PRF_X2.X2, PRF_E1.E1, PRF_E2.E2,
                 PRF_M1.M1, PRF_M2.M2, PRF_W.W,
                 MKT_X1.P1, MKT_X2.P2, MKT_PFX.PFX, MKT_L.PL,
                 MKT_K.PK, MKT_W.PW, I_CONS.CONS /;

*       Check the benchmark (again):

X1.L    =1;
X2.L    =1;
E2.L    =0;
M1.L    =0;
E1.L    =1;
M2.L    =1;
W.L     =1;

P1.L    =1;
P2.L    =1;
PFX.L   =1;
PK.L    =1;
PW.FX   =1;
PL.L    =1;

CONS.L  =200;

SOETARIFF.ITERLIM = 0;
```

```
SOLVE SOETARIFF USING MCP;

SOETARIFF.ITERLIM = 2000;
SOLVE SOETARIFF USING MCP;

* ccounterfactual experiment: free trade

TM2  = 0 ;
SOLVE SOETARIFF USING MCP;
```

8.4a Small open economy: modeling a quota as an endogenous tax equivalent

This model assumes the same benchmark data as the previous two models, but it assumes that there is a quota limiting imports.

The quota generates a gap between the foreign or world supply price and the domestic demand price for the “rationed” good.

This difference, often referred to as a quota rent (it is a form of Ricardian rent), must go to some agent.

In this model, we model this as a “tariff equivalent”.

The size of the tariff equivalent is endogenous, set so that the level of imports is less than or equal to the quota level.

We use an auxiliary variable, PQ, for an endogenous tax rate and a constraint equation to set the value of this “tax”.

The revenue from this tax is assigned to the representative consumer in the block for the import demand for M2.

The constraint equation (A_PQ) says to set the value of the tax PQ such that the activity level for imports (calibrated to be 1 initially) is less than or equal to 1.

```
A_PQ..    1 =G= M2;
```

And “tax” revenue (quota rent) is returned to the consumer.

```
CONS =E= 120*ENDOW*PL + 70*ENDOW*PK
              + 60*PFX*PM2*M2*PQ;
```

Units are chosen such that all domestic prices equal 1. The counterfactual experiment is to set remove the quota.

This is done with the statement PQ.FX = 0; since PQ is a variable, not a parameter.

An alternative way to do this is to declare a parameter and use it in place of the ‘1’ on the left-hand side of the constraint equation. Then the quota can be set at any level.

The value of PQ can be thought of as the tariff equivalent of the quota or the “shadow tariff”. To reintroduce the quota, we must free up PQ with two statements:

```
PQ.L = 0;
PQ.UP = +INF;
```

The second counterfactual is to triple the size of the economy.

```
$TITLE  M8-4a.GMS: Small open economy with a benchmark quota
* modeled as an endogenous tax rate

$ONTEXT
In this example, units are chosen such that all DOMESTIC prices
equal one initially.  Implied world price of import good
X2: P2 = 1/1.2

                 Production Sectors                          Consumer

Markets          |     X1       X2       E1       M2       W       CONS
-------------------------------------------------------------------------
P1               |    150                -50              -100
P2               |              40                 60     -100
PL               |   -100      -20                                  120
PK               |    -50      -20                                   70
PW               |                                     200     -200
PFX              |                      50     -50
Q (quota rent)   |                             -10                   10
-------------------------------------------------------------------------
$OFFTEXT

PARAMETERS
        PE2     Export price of good 2
        PM1     Import price of good 1
        PE1     Export price of good 1
```

```
         PM2     Import price of good 2
         ENDOW   Endowment multiplier (size of the economy);

PE1      = 1;
PM2      = 1 / (1.2);
PE2      = PM2 * 0.99;
PM1      = 1.01;
ENDOW    = 1;

NONNEGATIVE VARIABLES
    X1       Activity level for sector X1,
    X2       Activity level for sector X2,
    E1       Activity level for sector E1,
    E2       Activity level for sector E2,
    M1       Activity level for sector M1,
    M2       Activity level for sector M2,
    W        Activity level for sector W,
    P1       Price index for commodity X,
    P2       Price index for commodity Y,
    PL       Price index for primary factor L,
    PK       Price index for primary factor K,
    PW       Price index for welfare (expenditure function),
    PFX      Read exchange rate index,
    CONS     Income definition for CONS
    PQ       Quota shadow price (ad valorem tariff equivalent);
```

```
EQUATIONS
    PRF_X1  Zero profit for sector X1
    PRF_X2  Zero profit for sector X2
    PRF_E1  Zero profit for sector E1
    PRF_E2  Zero profit for sector E2
    PRF_M1  Zero profit for sector M1
    PRF_M2  Zero profit for sector M2
    PRF_W   Zero profit for sector W

    MKT_X1  Supply-demand balance for commodity X1
    MKT_X2  Supply-demand balance for commodity X2
    MKT_PFX Supply-demand balance for commodity PFX
    MKT_L   Supply-demand balance for primary factor L
    MKT_K   Supply-demand balance for primary factor L
    MKT_W   Supply-demand balance for aggregate demand

    I_CONS  Income definition for CONS
    A_PQ    Quota auxiliary (sets endogenous shadow tax PQ);

*       Zero profit conditions

PRF_X1..  150 * PL**(2/3) * PK**(1/3) =G= 150*P1;

PRF_X2..  40  * PL**(0.5) * PK**(0.5) =G= 40*P2;

PRF_E1..  50 * P1 =G= 50 * PFX * PE1;
```

```
PRF_E2..  60 * P2 =G= 60 * PFX * PE2;

PRF_M1..  50 * PFX * PM1 =G= 50 * P1;

PRF_M2..  60 * PFX * PM2 *(1+PQ) =G= 60*P2;

PRF_W..   200 * P1**0.5 * P2**0.5 =G= 200*PW;

*       Market clearance conditions

MKT_X1..  150*X1 + 50*M1 =G=  50*E1 + 100*W*PW/P1;

MKT_X2..  40*X2 + 60*M2 =G= 60*E2 + 100*W*PW/P2 ;

MKT_PFX.. 60*E2*PE2 + 50*E1*PE1 =G= 60*M2*PM2 + 50*PM1*M1;

MKT_W..   200*W =G= CONS / PW;

MKT_L..   120*ENDOW  =G= 100*X1*P1/PL + 20*X2*P2/PL;

MKT_K..   70*ENDOW   =G=  50*X1*P1/PK + 20*X2*P2/PK;

*       Income balance

I_CONS..  CONS =E= 120*ENDOW*PL + 70*ENDOW*PK + 60*PFX*PM2*M2*PQ;
```

```
A_PQ..    1 =G= M2;

MODEL SOE2 /PRF_X1.X1, PRF_X2.X2, PRF_E1.E1, PRF_E2.E2,
            PRF_M1.M1, PRF_M2.M2,
            PRF_W.W, MKT_X1.P1, MKT_X2.P2, MKT_PFX.PFX,
            MKT_L.PL, MKT_K.PK, MKT_W.PW,
            I_CONS.CONS, A_PQ.PQ /;

*       Check the benchmark (again):

X1.L    =1;
X2.L    =1;
E2.L    =0;
M1.L    =0;
E1.L    =1;
M2.L    =1;
W.L     =1;

P1.L    =1;
P2.L    =1;
PFX.L   =1;
PK.L    =1;
PW.FX   =1;
PL.L    =1;
```

```
CONS.L  =200;

PQ.L = 0.20;

SOE2.ITERLIM = 0;
SOLVE SOE2 USING MCP;

SOE2.ITERLIM = 2000;
SOLVE SOE2 USING MCP;

* counterfactual: fixed PQ = 0 to calculate free trade

PQ.FX = 0;
SOLVE SOE2 USING MCP;

* show that the quota becomes more restrictive
* as the economy grows

PQ.LO = 0;
PQ.UP = +INF;

ENDOW = 3;
SOLVE SOE2 USING MCP;

* show what would happen if there had been a fixed 0.20 tariff
* instead of the quota
```

```
PQ.FX = 0.20;
SOLVE SOE2 USING MCP;


PQ.LO = 0;
PQ.UP = +INF;

* show what would happen if the economy were smaller than the benchmark

ENDOW = 0.25;
SOLVE SOE2 USING MCP;
```

8.4b Small open economy: modeling a quota as a supply/demand for licenses

This is the same exercise as the previous M8.4a, except is shows an alternative way of modeling a quota or indeed any quantitative restriction.

This is to assume that the quota is enforced by licences sold to importers.

The government might auction them off, for example (which does happen in practice for certain licenses and permits).

We will assume that the government returns the receipt lump sum to the representative consumer.

Now, instead of an auxiliary variable and constraint equation, we have an added variable PLIC, the price of a license,

and a market clearing equation MKT_LIC which equates the exogenous supply of licences (SLIC) to the demand for them.

```
PARAMETER
   SLIC    Supply of import licenses for X2;

NON-NEGATIVE VARIABLE
   PLIC    Price of a license;

EQUATION
   MKT_LIC Market for import licenses;


PRF_M2..  60*PLIC + 60*PFX*PM2 =G= 60*P2;

MKT_LIC.. 60*SLIC =G= 60*M2;

I_CONS..  CONS =E= 120*ENDOW*PL + 70*ENDOW*PK + 60*PLIC*SLIC;
```

```
$TITLE  M8-4b: Small open economy with a benchmark quota
* modeled as supply/demand for import licenses

$ONTEXT
In this example, units are chosen such that all DOMESTIC
 prices equal one initially.
Implied world price of import good X2 P2 = 1/1.2


                 Production Sectors                        Consumer

Markets        |     X1       X2       E1       M2       W       CONS
---------------------------------------------------------------------
P1             |    150                -50              -100
P2             |              40                 60     -100
PL             |   -100      -20                                 120
PK             |    -50      -20                                  70
PW             |                                     200     -200
PFX            |                       50     -50
PLIC           |                              -10                 10
---------------------------------------------------------------------

$OFFTEXT

PARAMETERS
        PE2     Export price of good 2
```

```
         PM1      Import price of good 1
         PE1      Export price of good 1
         PM2      Import price of good 2
         SLIC     Supply of import licenses for X2 (M2)
         ENDOW    Endowment multiplier (size of the economy);

PE1      = 1;
PM2      = 1 / (1.2);
PE2      = PM2 * 0.99;
PM1      = 1.01;
SLIC     = 1;
ENDOW    = 1;

NONNEGATIVE VARIABLES
    X1       Activity level for sector X1,
    X2       Activity level for sector X2,
    E1       Activity level for sector E1,
    E2       Activity level for sector E2,
    M1       Activity level for sector M1,
    M2       Activity level for sector M2,
    W        Activity level for sector W,
    P1       Price index for commodity X,
    P2       Price index for commodity Y,
    PL       Price index for primary factor L,
    PK       Price index for primary factor K,
    PW       Price index for welfare (expenditure function),
```

```
    PFX     Read exchange rate index,
    CONS    Income definition for CONS
    PLIC    Price of a license (ad valorem tariff equivalent);

EQUATIONS
    PRF_X1  Zero profit for sector X1
    PRF_X2  Zero profit for sector X2
    PRF_E1  Zero profit for sector E1
    PRF_E2  Zero profit for sector E2
    PRF_M1  Zero profit for sector M1
    PRF_M2  Zero profit for sector M2
    PRF_W   Zero profit for sector W

    MKT_X1  Supply-demand balance for commodity X1
    MKT_X2  Supply-demand balance for commodity X2
    MKT_PFX Supply-demand balance for commodity PFX
    MKT_L   Supply-demand balance for primary factor L
    MKT_K   Supply-demand balance for primary factor L
    MKT_W   Supply-demand balance for aggregate demand

    I_CONS  Income definition for CONS
    MKT_LIC Market for import licenses;

*       Zero profit conditions

PRF_X1..    150* PL**(2/3) * PK**(1/3) =G= 150*P1;
```

```
PRF_X2..    40* PL**(0.5) * PK**(0.5) =G= 40*P2;

PRF_E1..    50*P1 =G= 50*PFX*PE1;

PRF_E2..    60*P2 =G= 60*PFX*PE2;

PRF_M1..    50*PFX * PM1 =G= 50*P1;

PRF_M2..    60*PLIC + 60*PFX*PM2 =G= 60*P2;

PRF_W..     200*P1**0.5 * P2**0.5 =G= 200*PW;

*       Market clearance conditions

MKT_X1..    150*X1 + 50*M1 =G=  50*E1 + 100*W *PW/P1;

MKT_X2..    40*X2 + 60*M2 =G= 60*E2 + 100*W*PW/P2 ;

MKT_PFX..   60*E2*PE2 + 50*E1*PE1 =G= 60*M2*PM2 + 50*PM1*M1;

MKT_W..     200*W =G= CONS / PW;

MKT_L..     120*ENDOW  =G= 100*X1*P1/PL + 20*X2*P2/PL;

MKT_K..     70*ENDOW   =G=  50*X1*P1/PK + 20*X2*P2/PK;
```

```
MKT_LIC..   60*SLIC =G= 60*M2;

*       Income balance

I_CONS..    CONS =E= 120*ENDOW*PL + 70*ENDOW*PK + 60*PLIC*SLIC;


MODEL ALGEBRAIC /PRF_X1.X1, PRF_X2.X2, PRF_E1.E1, PRF_E2.E2,
                 PRF_M1.M1, PRF_M2.M2,
                 PRF_W.W, MKT_X1.P1, MKT_X2.P2, MKT_PFX.PFX,
                 MKT_L.PL, MKT_K.PK, MKT_W.PW,
                 MKT_LIC.PLIC, I_CONS.CONS/;

*       Check the benchmark:

X1.L    =1;
X2.L    =1;
E2.L    =0;
M1.L    =0;
E1.L    =1;
M2.L    =1;
W.L     =1;

P1.L    =1;
P2.L    =1;
```

```
PFX.L   =1;
PK.L    =1;
PW.FX   =1;
PL.L    =1;

CONS.L  =200;

PLIC.L = 1/6;

ALGEBRAIC.ITERLIM = 0;
SOLVE ALGEBRAIC USING MCP;

ALGEBRAIC.ITERLIM = 2000;
SOLVE ALGEBRAIC USING MCP;

* show what happens if the supply of licenses is greatly expanded

SLIC = 5;
SOLVE ALGEBRAIC USING MCP;

* show that the quota becomes more restrictive as
* the economy grows

SLIC = 1;
ENDOW = 3;
SOLVE ALGEBRAIC USING MCP;
```